
News  

Meetings and events 
18th Annual Meeting of the Society for the Neural Control of Movement 
Naples Beach Hotel 
Naples, Florida  
April 29th – May 4th 2008 

Vision Sciences Society Annual Meeting 
Naples Grande Hotel 
Naples, Florida  
May 9th – 14th 2008 

2nd Annual Canadian Neuroscience Meeting 
Hilton Bonaventure Hotel 
Montreal, Canada  
May 25th – 28th 2008 

International Society of Electrophysiology and Kinesiology (ISEK) 
Sheraton on the Falls Hotel 
Niagara Falls, Canada  
June 18th – 21st 2008 

 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.08 Signal version 4.01 

Spike2 version 5.19 Signal version 3.10 

Spike2 demo Signal demo 

Spike2   
 

Q. I am interested in analysing baroreceptor sensitivity, specifically the result of 
dividing systolic blood pressure by the R-R interval. Is this possible in Spike2? 

A. Using the Active Cursors and Measure to channel functions in Spike2 we can 
create new data channels in the file holding the measurements of the R-R 
interval and the systolic blood pressure value. These could then be combined in 
a virtual channel to display the result of systolic BP divided by R-R interval. 

http://ncm-society.org/index.php?page=annual_meeting
http://www.visionsciences.org/meeting.html
http://www.can-acn2008.org/
https://www.isek2008.ca/default.asp
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml
CED
We also have some example scripts available for BRS analysis. Please contact us for details



 
Example ECG and BP data 

R-R Interval channel 
 In this example Cursor 0 is set up to find the systolic peak and Cursor 1 is set up 

to find the previous systolic peak in the BP channel. We are going to use the 
measured interval between systolic peaks as our R-R interval to save time in 
setting up cursors although, for the purists amongst you, it is of course possible 
to search for the R wave peaks in the ECG channel to get the true R-R interval. 

Create a new Cursor 1 from the Cursor menu and then select Active modes. For 
the example above, we can set Cursor 0 to find Peaks in the BP channel that 
have an amplitude change of 25 mmHg. This is a large enough amplitude 
change in the example data to detect the systolic peak, without detecting the 
dicrotic notch area. Cursor 1 is set up to find peaks with the same amplitude 
value, but is set to search from Cursor 0 to Cursor 0 minus 1 second, thereby 
finding the previous systolic peak in the trace. You can check that the cursors are 
finding the correct features by holding down Ctrl + Shift and using the left 
and right arrow keys to step to the previous and next peaks. 

   
Cursor active modes for interval detection 

 Now you can use the Analysis menu->Measurements->Data channel command 
to step the cursors through the data and write measurements to a data channel. 
The example below creates a new RealMark data channel holding the 
measurements of the Time difference between Cursor 0 and Cursor 1 to give us 
our R-R interval measurement. 

CED
You can also create cursors using the Ctrl + n shortcut where n is a number between 0-9 and access the active modes by right-clicking on a cursor



 
Measure to channel dialog for intervals 

Click New to create the measurement channel and then Process in the second 
dialog to take the measurements to channel m1 (channel 401). 

 
  

Process dialog and R-R interval channel m1 (channel 401) 

Systolic BP channel 
 The next step is to create a second channel holding the value of the systolic 

blood pressure from the example BP trace. To do this, select Measurements and 
then Data channel again from the Analysis menu. For the systolic value, change 
the Y-measurement setting to Value at point of Cursor 0 and create the new 
channel by clicking on New and then Process, as before. This creates channel 
m2 (channel 402) holding the systolic values. 

Systolic BP / RR Interval 
 Now we have our measurements, we can use a Virtual Channel to display the 

systolic values divided by the R-R interval values. The Analysis menu Virtual 
Channels->Create New Channel command generates channel v1 into which we 
can combine the data from both of our RealMark channels. The RM() function 
accesses RealMark data and converts it to a waveform. The expression below 
converts both channels to a waveform and divides the systolic channel, m2 by 
the interval channel m1 to show the results in channel v1. 

 

 

 

 

 

CED
You can ‘build’ virtual channels interactively by following the arrow button in the dialog (highlighted to the left) and selecting from a list of available expressions



 
Virtual channel v1 showing the result  

The measurement channels in the above example were created as memory 
channels, so are not saved when the data file closes. You can save memory 
channels from the Memory Buffer -> Write to Channel command from the 
Analysis menu, or create the measurement channels as permanent data 
channels in the Measure to Channel dialog. Further examples of using active 
cursors, measurement features and virtual channels can be found in the on-line 
help and in previous eNewsletters. 

 

  
 

Q. Does Signal support the A-M Systems 2400 amplifier for telegraph outputs? 

A. Signal can configure up to four ports whose scaling during sampling depends on 
voltage values received via telegraph outputs. You can generate voltage values 
and corresponding gain settings for a list of known amplifiers from the 
appropriate button in the dialog. 

 
Standard telegraph dialog  

 If your amplifier is not included then please contact us and we will try to add it. 
You can use the Volts and Gain controls to directly enter a voltage value and 
corresponding gain and Add it to a manual list of telegraph settings. 

 Our thanks to John Dempster at the University of Strathclyde for providing much 
of the initial amplifier information.  

CED
Signal also offers telegraph support for the MultiClamp 700 series. This replaces the standard telegraph configuration and should be selected during installation if required



Scripts: Spike2  
Q. I want to control the sequencer output on-line using a script so that I can change 

pulse amplitude values during an experiment. Do you have any examples of 
using a script and sequence in this way? 

A. The attached script CursorPulseAmp.s2s, along with the associated sequence 
file, CursorPulseAmp.pls is an example of updating a running sequence file 
during sampling with an interactive script. The attached sampling configuration, 
CursorPulseAmp.s2c can be used to visualize the pulse outputs by 
connecting DAC 0 to ADC 0 on the 1401. Load the CursorPulseAmp.pls text 
sequence from the sequencer tab of the sampling configuration and click OK 
before running the script. The script opens a new data file ready to sample and 
creates a toolbar to start and stop sampling. During sampling the horizontal 
cursor can be dragged up and down to control the amplitude of the output pulse 
train. 

 

Scripts: Signal  
Q. I would like to create new data files that contain only frames of a certain state 

from existing experiment files. Is there any way to do this in Signal? 

A. The attached script, Export&Merge.sgs, can be used to create a new data file 
by merging together frames from existing files based on frame state values. You 
can open files and select the state number to use from the script toolbar. The 
frames must all have the same number of data points to be included in the newly 
created file. 

Did you know…? 
You can now sample for much longer run times with the same sample rates when using 
burst mode sampling in Spike2. Burst mode is supported by the Micro1401 mkII, 
Power1401 and Power1401 mkII. 

Recent questions   
Q. Is the any way to merge files together in Spike2? Ideally I would like to be able to 

process a folder of files recorded at different times into a single data file. 

A. The script MergeFiles.s2s, included in the Spike6\scripts folder for all 
new Spike2 6.08 installations, allows you to combine multiple data files and copy 
and paste channel data between files. This script is a greatly improved and more 
comprehensive version than the script included in eNewsletter 32 and has three 
basic modes of operation controlled via a script toolbar: 

 
This mode allows the user to create a new destination file holding the data from a 
directory of source files linked end to end in order of creation date. Alternatively it 
can be used to add data from the source files into new channels in the 
destination file so that all channels share the same time range. This mode of 
processing would usually be used for merging files that were recorded with the 
same sampling configuration. 

 
This mode allows the user to copy and paste data between channels and data 
files. You can select the destination file for the paste from an existing open file or 
create a new file to hold the copied data. Like the Batch Process mode, this 
works best with channels and files that share the same sampling rates and time 
resolution. If the source data does not match the destination in these respects 
then the data is represented as best it can using cubic splining. 

The Copy/Paste mode offers greater flexibility than Batch Process as the user 
can copy groups of channels in a time range and paste them to the destination 
file at a specified time, overwriting pre-existing data if required. 

Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$CursorPulseAmp|Example script to show control of sequencer output interactively

'It contains toolbar to open a new data file for sampling and
'start and stop sampling. Should be used with the associated sequencer file
'CursorPulseAmp.pls and a suitable sampling configuration containing 1 waveform
'channel sampled at 5kHz. Connect DAC0 to waveform input.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED     18/04/08

var data%;                      'Handle of new data file (Spike2 sets it to 0)
var sTime;                      'The last time we looked at the idle routine
var pLev;                       'Previous horizontal cursor position 
var fullMax,fullMin;            'Full input range of channel based on scale and offset
var chan% := 1;

HideAll();                      'Hide all unwanted bars
ToolbarVisible(1);              'Make toolbar visible always
New%();                         'Set up new sampling window
DoToolbar();                    'Do the toolbar
RestoreAll();                   'Restore all bars
Halt;

func New%()                     'New sampling window

if data%>0 then                 'If there is already a data view open...
  View(data%);                  '...close it
  FileClose();                  'No point setting data% to 0...
endif;

data%:=FileNew(0,1);            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
Window(0,0,100,100);             'Make data window in top bit of screen
XRange(0,10);
fullMax := (ChanScale(chan%)*5) + ChanOffset(chan%) - (1/(Pow(2,32)*ChanScale(chan%)));         'Get maximum allowed range (1 bit less than 5V)
fullMin := (ChanScale(chan%)*-5) + ChanOffset(chan%);                                           'Get minumum allowed range
HCursorDelete(-1);              'Clear any existing cursors
HCursorNew(chan%,1);          'Set new cursor
pLev := HCursor(1);
SetAmp(HCursor(1));
FrontView(data%);               'Bring the data view to the front
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(2,1);             'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(0,"",Idle%);         'Call Idle%() whenever there is free time
ToolbarSet(1,"&Quit",Quit%);    'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarEnable(3,0);             'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()                    'If "Quit" is pressed
SampleStop();                   'Stop sampling
return 0;                       'leave toolbar
end;

func Start%()                   'If "Start" is pressed
sTime := 0;                     'set start of anaylsis time
SampleStart();                  'Start sampling
SampleKey("s");                 'Start pulses
ToolbarEnable(4,0);             'Disable "New file" button
ToolbarEnable(3,1);             'Enable "Sample stop" button
ToolbarEnable(2,0);             'Disable "Sample start" button
ToolbarEnable(1,0);             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                       'Stay with toolbar
end;

func Stop%()                    'If "Stop" is pressed
SampleStop();                   'Stop sampling
if (data%>0) and                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);             'Enable "New file" button
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(1,1);             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                       'Stay in toolbar
end;

func Idle%()                    'Idle routine runs when system has time
var eTime;
if ViewKind(data%)=0 then       'This checks that the data view exists...
    View(data%);                 '...as this gives an error if it doesn't
        
    if SampleStatus() = 2 then    'if currently sampling
        if HCursorChan(1) = chan% then
            if HCursor(1) <> pLev then      'If horizontal cursor level changed
                if HCursor(1) >= fullMin and HCursor(1) <= fullMax then
                    SetAmp(HCursor(1));         'Calculate and pass new level to sequencer
                    pLev := HCursor(1);         'Update previous level variable
                endif;
            endif;
        endif;
    endif;
endif;
return 1;                       'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end


Proc SetAmp(amp);          'Procedure to calculate pulse amplitude in sequencer units and pass to sequencer variable
var seqAmp%;               'Sequencer amplitude as 32 bit integer
var curAmp;                'value on channel as a proportion of the ADC range 
curAmp := (amp - ChanOffset(chan%)) / (ChanScale(chan%) * 10);          
seqAmp% := curAmp * (Pow(2,32))+1;
SampleSeqVar(1,seqAmp%);
end;



CED
CursorPulseAmp.s2s




CED
CursorPulseAmp.pls


david
CursorPulseAmp.S2C


'$Export&Merge|Script to create a new data file from frames of existing files
'Gets frames with the same state in a file and adds these into created file
'The files must have the same number of data points to be included

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED     18/04/08


var file%;						'Source data file
var view%;						'Created memory view
var frame%;						'Source frame
var fcoun%;						'Number of frames in created file
var fpoints%,vpoints%;			'Number of points in source and created channels
var script%;					'Script window
var numframe%;					'Number of frames in selected file
var numChan%;
var state%;						'frame state chosen
var setState%;                  'State value o write to new file
var exportstate%;				'State of current frame	
var framenum%;					'Number of current frame in selected file
var neworadd%;					'Add to existing or create new file

script%:=App(3);				'Get handle of script window
View(script%).WindowVisible(0);	'Hide script window
ToolbarSet(1,"Quit");			'Create toolbar
ToolbarSet(3,"Select first file",NFile%);			'Link to function
Toolbar("Create new file from existing data",160);	'Show toolbar

Func NFile%()					'Function to select initial source file
file%:=FileOpen("",0,0);		'Open file
Frame(1);
numChan%:=ChanCount(-4);
WindowVisible(3);				'Maximise view
ToolbarClear(3);				'Clear toolbar button
ToolbarSet(3,"New file",NewFile%);				'New toolbar button linked to function
ToolbarSet(6,"Separate state",Getstate%);					'Link to function
ToolbarSet(7,"Delete frame",Del%);				'Link to function
ToolbarEnable(7,0);								'Disable button
Return 1;
End;


Func NewFile%()					'Function to select new source file
View(file%);					
FileClose();					'Close existing source file
file%:=FileOpen("",0,0);		'Open new file
Window(50,0,100,100);			'Position window
FrontView(file%);				'Bring to front
Return 1;
End;

Func Getstate%()						'Function to add channel of data to created view
var okcan%;						'Check cancel button not pressed on dialog box
var doChanExport%;
if view%>0 then
	neworadd%:=Query("Add to existing file or create new file","Add","New");
	if neworadd%=0 then
		View(view%);
		FileClose();
		fcoun%:=0;
	endif;
endif;				
DlgCreate("State selection");	'Create dialog box
DlgInteger(1,"Select state to export (-1 = all frames)",-1,100);	'Select channel
okcan%:=DlgShow(state%);				
If okcan% = 1 then					'If ok pressed 
	View(file%).Frame(1);			'Go to first frame 
	framenum%:=1;
    if state% > -1 then
	exportState% := FrameState(-2);
	While exportstate%<>state% do	'If state of current frame not selected state
		framenum%:=framenum%+1;		'Go to next frame
		Frame(framenum%);	
		exportstate%:=FrameState(-2);
            Wend;
    endif;
    
	endif;
	if fcoun% =0 then				'If first frame to be included
		view%:=SetCopy(-1,"New file",0);		'Create average view
	
		for doChanExport%:=1 to numChan% do
			ArrConst(View(view%,doChanExport%).[],View(file%,doChanExport%).[]);	'Copy data from source to created frame
			View(view%).FrameState(-2,state%);
		next;
		Window(0,0,50,100);			'Position window
		FrontView(view%);
		fpoints%:=ChanPoints(1);	'Count points in a channel
		View(file%).Window(50,0,100,100);	'Position window
		fcoun%:=fcoun% + 1;			'Stop script repeating this section
	
		framenum%:=framenum%+1;
		View(file%);				
		vpoints%:=ChanPoints(1);	'Get number of points in frame
		If vpoints%<>fpoints% then		'If not the same as created file
			Message("Data files incompatible\nPlease try another");	'Tell user to select another file
			Return 1;
		Endif;
		View(file%);				
		numframe%:=FrameCount();			'Count frames in file

		Repeat								'Loop to extract all frames of same state
			View(file%);
			Frame(framenum%);				'Go to last frame
			exportstate%:=FrameState(-2);	'Get state 
			if exportstate% = state% or state% = -1 then		'If selected state or all states selected
                    View(view%);
                    AppendFrame();				'Append empty frame to end of created file
                    fcoun%:=FrameCount();		'Get number of frames in created file
                    Frame(fcoun%);				'Go to last frame
                    for doChanExport%:=1 to numChan% do
                        ArrConst(View(view%,doChanExport%).[],View(file%,doChanExport%).[]);	'Copy data from source to created frame
                    next;
                if state% = -1 then
                    setState% := exportState%;
                else
                    setState% := state%;
                endif;
                View(view%).FrameState(-2,setState%);
			endif;
			framenum%:=framenum%+1;
			Until framenum%>numframe%;
		ToolbarEnable(7,1);			'Enable button	
	else
    
        View(file%);				
		vpoints%:=ChanPoints(1);	'Get number of points in frame
		If vpoints%<>fpoints% then		'If not the same as created file
			Message("Data files incompatible\nPlease try another");	'Tell user to select another file
			Return 1;
		Endif;
		View(file%);				
		numframe%:=FrameCount();

		Repeat								'Loop to extract all frames of same state
			View(file%);
			Frame(framenum%);
			exportstate%:=FrameState(-2);
			if exportstate%=state% or state% = -1 then		'If selected state
				View(view%);
				AppendFrame();				'Append empty frame to end of created file
				fcoun%:=FrameCount();		'Get number of frames in created file
				Frame(fcoun%);				'Go to last frame
				for doChanExport%:=1 to numChan% do
					ArrConst(View(view%,doChanExport%).[],View(file%,doChanExport%).[]);	'Copy data from source to created frame
					next;
                            if state% = -1 then
                    setState% := exportState%;
                else
                    setState% := state%;
            endif;
            View(view%).FrameState(-2,setState%);
			endif;
			framenum%:=framenum%+1;
			Until framenum%>numframe%;
		ToolbarEnable(7,1);			'Enable button	
	Endif;
Return 1;
End;



Func Del%()							'Function to delete frame
View(view%);		
DeleteFrame();						'Delete selected frame	
Return 1;
End;

CED
Export & merge.sgs



 
This mode can be used to delete unwanted sections of data, such as artefacts, 
from single or multiple channels in a file. Deleted sections of waveform data can 
be replaced with a baseline at zero or a gap in the data. The user can also select 
to close gaps to defined time durations or join data end to end with no break in 
the recording. 

 If you have any comments or suggestions when using the MergeFiles.s2s 
script, then please contact us. 

  

User group      
If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

